Homogeneous Solution of the Einstein-Maxwell Equations

A

Synopsis

We exhibit and analyze a homogeneous spacetime whose source is a pure radiation electromagnetic field [1]. It was previously believed that this spacetime is the sole example of a homogeneous pure radiation solution of the Einstein equations which admits no electromagnetic field (see [2] and references therein). Here we correct this error in the literature by explicitly displaying the electromagnetic source. This result implies that all homogeneous pure radiation spacetimes satisfy the Einstein-Maxwell equations.

In this worksheet we shall

1. Display the metric and electromagnetic field and verify that the Einstein and Maxwell equations are satisfied.
2. Exhibit some geometric properties of the solution: the electromagnetic field is null, the Petrov type is N, and the spacetime is a non-reductive homogeneous space.

1. The Einstein-Maxwell solution

Load in the required packages.

```
> with(DifferentialGeometry): with(Tensor): with(GroupActions):
  with(LieAlgebras): with(Library):
```

Define a chart and construct the metric. This is (12.36) in "Exact Solutions to Einstein's Field Equations" by Stephani, et al. We can obtain this metric from the DG library using the Retrieve command.

```
> DGsetup([u,v,x,y], M);
```

$$\textit{frame name: M} \tag{1.1}$$

```
M > g := Retrieve("Stephani", 1, [12, 36, 1], manifoldname = M,
    output = ["Fields"])[1];
```

$$g := -2\,e^{2\,\_\rho\,x}\,du \otimes du + du \otimes dv + dv \otimes du + dx \otimes dx + dy \otimes dy \tag{1.2}$$

Redefine the parameter name.

```
M > g := eval(g, _rho = rho);
```

$$g := -2\,e^{2\,\rho\,x}\,du \otimes du + du \otimes dv + dv \otimes du + dx \otimes dx + dy \otimes dy \tag{1.3}$$

Compute the Einstein tensor. It is of the form $G = \Phi^2\ \partial_v \otimes \partial_v$ corresponding to an energy-momentum tensor of pure radiation type in terms of the null vector field ∂_v.

```
M > G := EinsteinTensor(g);
```

$$G := 4\,\rho^2\,e^{2\,\rho\,x}\,D\_v \otimes D\_v \tag{1.4}$$

The following 2-form is the electromagnetic field strength. The function $f(u)$ is arbitrary.

```
M > F := evalDG(2*exp(rho*x)*rho*cos(rho*y + f(u))*du &w dx - 2*
    exp(rho*x)*rho*sin(rho*y + f(u))*du &w dy);
```

$$F := 2\,e^{\rho\,x}\,\rho\,\cos(\rho\,y + f(u))\,du \wedge dx - 2\,e^{\rho\,x}\,\rho\,\sin(\rho\,y + f(u))\,du \wedge dy \tag{1.5}$$

We verify the Einstein equations are satisfied by computing the electromagnetic energy-momentum tensor,

```
M > T := EnergyMomentumTensor("Electromagnetic", g, F);
```

$$T := 4\,\rho^2\,e^{2\,\rho\,x}\,D\_v \otimes D\_v \tag{1.6}$$

and then verifying that it equals the Einstein tensor:

```
M > evalDG(G - T);
```

$$0 \tag{1.7}$$

Next we verify that the 2-form F satisfies the source-free Maxwell equations in the spacetime defined by g.

```
M > MatterFieldEquations("Electromagnetic", g, F);
```

$$0\,D\_u,\ 0\,du \wedge dv \wedge dx \tag{1.8}$$

2. Properties of the solution

Here we exhibit some geometrical properties of this electrovac spacetime.

The electromagnetic field is null.

We begin by establishing the electromagnetic field is null: its two scalar invariants vanish. The first invariant is the complete contraction of F with itself.

```
M > TensorInnerProduct(g, F, F);
```

$$0 \tag{2.1}$$

The second invariant is the Hodge dual of F and its complete contraction with F.

```
M > Fdual := HodgeStar(g, F, detmetric = -1);
```

$$Fdual := 2\,e^{\rho\,x}\,\rho\,\sin(\rho\,y + f(u))\,du \wedge dx + 2\,e^{\rho\,x}\,\rho\,\cos(\rho\,y + f(u))\,du \wedge dy \tag{2.2}$$

```
M > TensorInnerProduct(g, F, Fdual);
```

$$0 \tag{2.3}$$

The Petrov type is N.

The Petrov type is easily computed.

```
M > PetrovType(g);
```

"N" **(2.4)**

There are 2 covariantly constant vector fields.

The space of covariantly constant vector fields is two-dimensional.

```
M > CovariantlyConstantTensors(g, [D_u, D_v, D_x, D_y]);
```

$$[D\_y, D\_v] \quad \textbf{(2.5)}$$

The isometry group is 5 dimensional and the electromagnetic field is non-inheriting.

We find a basis for the Killing vectors, which generate a five dimensional group of isometries.

```
M > KV := KillingVectors(g);
```

$$KV := [-y\, D\_v + u\, D\_y, D\_y, -\rho\, u\, D\_u + \rho\, v\, D\_v + D\_x, D\_u, D\_v] \quad \textbf{(2.6)}$$

We take the Lie derivative of the metric to confirm that these vector fields are in fact Killing vector fields.

```
M > LieDerivative(KV, g);
```

$$[0\, du \otimes du, 0\, du \otimes du, 0\, du \otimes du, 0\, du \otimes du, 0\, du \otimes du] \quad \textbf{(2.7)}$$

The electromagnetic field is not invariant under the isometry group of the spacetime. For example,

```
M > LieDerivative(KV[2], F);
```

$$-2\, e^{\rho x} \rho^2 \sin(\rho\, y + f(u))\, du \wedge dx - 2\, e^{\rho x} \rho^2 \cos(\rho\, y + f(u))\, du \wedge dy \quad \textbf{(2.8)}$$

The spacetime is a non-reductive homogeneous space.

The command SubspaceType computes the dimension and metric-induced signature of the subspace of the tangent space which is spanned by the Killing vectors at a given point. We see that the Killing vectors span the tangent space, so that the group they generate is transitive and the spacetime is a homogeneous space.

```
M > SubspaceType(g, KV, [u = u0, v = v0, x = x0, y = y0])
    assuming y0::real, y0 <> 0;
```

4, "PseudoRiemannian" **(2.9)**

Next we compute the Lie algebra of the isometry group from the brackets of the Killing vector fields and initialize the algebra (calling it "alg").

```
M > KValg := LieAlgebraData(KV, alg);
```

$KValg := [[e1, e2] = e5, [e1, e3] = \rho\, e1, [e1, e4] = -e2, [e3, e4] = \rho\, e4, [e3, e5] = -\rho\, e5]$ **(2.10)**

```
M > DGsetup(KValg):
```

The command IsotropySubalgebra finds a basis for the generators of the subgroup of the isometry group which fixes a given point. Since the space is homogeneous, we can just pick the origin of the chart without loss of generality. The output option exhibits the isotropy subalgebra basis in terms of the chosen basis for alg.

```
M > h:=IsotropySubalgebra(KV, [u = 0, v = 0, x = 0, y = 0],
    output=[alg]);
```

$$h := [e1] \tag{2.11}$$

A homogeneous space G/H with transitive group G and corresponding Lie algebra $\mathfrak{g}$ is called *reductive* if the isotropy subalgebra $\mathfrak{h}$ admits an invariant complement m in $\mathfrak{g}$, that is, $\mathfrak{g} = \mathfrak{h} + m$, and $[\mathfrak{h}, m] \subseteq m$. In this case $(\mathfrak{h}, m)$ is called a *reductive pair.* We now determine if this electrovac solution is a reductive homogeneous space. First construct the general form of the complement of **h** with parameters (if any) labeled by t.

```
L1 > m := ComplementaryBasis(h, [e1, e2, e3, e4, e5], t);
```

$$m := [t1\, e1 + e2, t2\, e1 + e3, t3\, e1 + e4, t4\, e1 + e5], \{t1, t2, t3, t4\} \tag{2.12}$$

Now use the Query command to determine if this 4-parameter family of complements includes an $\mathfrak{h}$ - invariant complement.

```
alg > LieAlgebras:-Query(h, m, "ReductivePair")[1];
```

$$\mathit{false} \tag{2.13}$$

This spacetime is therefore a non-reductive homogeneous space.

Commands Illustrated

ComplementaryBasis, CovariantlyConstantTensors, EinsteinTensor, EnergyMomentumTensor, HodgeStar, IsotropySubalgebra, KillingVectors, LieAlgebraData, LieDerivative, MatterFieldEquations, PetrovType, Query, SubspaceType, TensorInnerProduct,

References

1. Torre, C.G., "All homogeneous pure radiation spacetimes satisfy the Einstein-Maxwell equations", *Class. Quant. Grav.* **29** (2012) 077001.
2. Stephani, H. Kramer, D MacCallum, M. Hoenselaers, C., and Herlt, E. *Exact Solutions to Einstein's Field Equations*. 2nd ed. (Cambridge Monographs on Mathematical Physics, 2003) .

Release Notes

• The illustrated commands are available in Maple 13 and subsequent releases.

Author

C. G. Torre
Department of Physics

Utah State University
July 18, 2012